

<requete> ::= <create_table>

| <insert_into>

| <select>

| <update>

| <delete>

| <drop_table>

<select> ::= SELECT <liste_champs> FROM <nom_table> ;

| SELECT <liste_champs> FROM <nom_table> WHERE <condition> ;

<liste_champs> ::= *

| <liste_noms_champs>

<liste_noms_champs> ::= <nom_champ>

| <nom_champ> , <liste_noms_champs>

<create_table> ::= CREATE TABLE <nom_table> (<liste_definitions>) ;

<liste_definitions> ::= <definition_colonne>

| <definition_colonne> , <liste_definitions>

<definition_colonne> ::= <nom_champ> <type_donnee>

<type_donnee> ::= INT

| FLOAT

| BOOL

| VARCHAR (<entier>)

<insert_into> ::= INSERT INTO <nom_table> VALUES (<liste_valeurs>) ;

| INSERT INTO <nom_table> (<liste_noms_champs>) VALUES (<liste_valeurs>) ;

<liste_valeurs> ::= <valeur>

| <valeur> , <liste_valeurs>

<update> ::= UPDATE <nom_table> SET <liste_affectations> WHERE <condition> ;

<liste_affectations> ::= <affectation>

| <affectation> , <liste_affectations>

<affectation> ::= <nom_champ> = <valeur>

<delete> ::= DELETE FROM <nom_table> ;

| DELETE FROM <nom_table> WHERE <condition> ;

<drop_table> ::= DROP TABLE <nom_table> ;

<condition> ::= <condition_ou>

<condition_ou> ::= <condition_et>
| <condition_et> OR <condition_ou>

<condition_et> ::= <condition_not>
| <condition_not> AND <condition_et>

<condition_not> ::= NOT <condition_simple>
| <condition_simple>

<condition_simple> ::= <operande> <opérateur_comp> <operande>
| (<condition>)

<operande> ::= <nom_champ>
| <valeur>

<opérateur_comp> ::= = | != | < | > | <= | >=

<nom_table> ::= IDENTIFICATEUR
<nom_champ> ::= IDENTIFICATEUR

<valeur> ::= <entier>
| <reel>
| <chaîne>
| <booleen>

<entier> ::= [0-9]+
<reel> ::= [0-9]+.[0-9]+
<chaîne> ::= '...' | "..."
<booleen> ::= TRUE | FALSE